

ColorPAL (#28380) Arduino Demo




Please note: This demo was created to support the 2013 National microMedic Contest kits, which are no longer available.

The ColorPAL is a sensor for detecting the color of a surface. This tutorial, supporting the 2013 National microMedic Contest, connects a ColorPal to a BOE Shield for Arduino for reading the color of surfaces, such as a pH test strip.

About the ColorPAL

The ColorPAL alternately activates each one of three internal colored light sources -- red, green, and blue -- shining those colors onto the surface. It then detects the amount of illumination that is reflected back for each of these colors.

The relative amounts of red, green, and blue light that is reflected indicates the color of the object. A red object, for example, will reflect much more red than a green object. The sensor interprets the values of red, green, and blue to evaluate the shade of many thousands of different colors.




The ColorPAL cannot be used to determine the color of an object that emits its own light. You cannot use it, for instance, to test the color displayed on your computer monitor.

Another Color Sensor Option

If you require a sensor that can detect the color of objects that give off their own light, look at the Parallax TCS3200-DB Color Sensor (#28302). The TCS3200-DB sensor also incorporates a lens, so it has a narrower field of view (area of detection).

Connections

To connect the ColorPAL to the BOE Shield for Arduino Shield, attach its three leads as shown in the figure below.



Programming

To use this example, upload the example sketch to your Arduino, then open the Serial Monitor window. Ensure that the Baud Rate is set at 9600. Hold the detector window of the ColorPAL over a surface, such as a piece of colored paper.

```
/* ColorPal Sensor Example for Arduino
   Author: Martin Heermance, with some assistance from Gordon McComb
   This program drives the Parallax ColorPAL color sensor and provides
   serial RGB data in a format compatible with the PC-hosted
   TCS230_ColorPAL_match.exe color matching program.
*/

#include <SoftwareSerial.h>

const int sio = 2;           // ColorPAL connected to pin 2
const int unused = 255;      // Non-existent pin # for SoftwareSerial
const int sioBaud = 4800;
const int waitDelay = 200;

// Received RGB values from ColorPAL
int red;
int grn;
int blu;
```

```

// Set up two software serials on the same pin.
SoftwareSerial serin(sio, unused);
SoftwareSerial serout(unused, sio);

void setup() {
  Serial.begin(9600);
  reset(); // Send reset to ColorPal
  serout.begin(sioBaud);
  pinMode(sio, OUTPUT);
  serout.print("= (00 $ m) !"); // Loop print values, see ColorPAL
documentation
  serout.end(); // Discontinue serial port for transmitting

  serin.begin(sioBaud); // Set up serial port for receiving
  pinMode(sio, INPUT);
}

void loop() {
  readData();
}

// Reset ColorPAL; see ColorPAL documentation for sequence
void reset() {
  delay(200);
  pinMode(sio, OUTPUT);
  digitalWrite(sio, LOW);
  pinMode(sio, INPUT);
  while (digitalRead(sio) != HIGH);
  pinMode(sio, OUTPUT);
  digitalWrite(sio, LOW);
  delay(80);
  pinMode(sio, INPUT);
  delay(waitDelay);
}

void readData() {
  char buffer[32];

  if (serin.available() > 0) {
    // Wait for a $ character, then read three 3 digit hex numbers
    buffer[0] = serin.read();
    if (buffer[0] == '$') {
      for(int i = 0; i < 9; i++) {
        while (serin.available() == 0); // Wait for next input character
        buffer[i] = serin.read();
        if (buffer[i] == '$') // Return early if $ character
          encountered
            return;
      }
      parseAndPrint(buffer);
      delay(10);
    }
  }
}

// Parse the hex data into integers
void parseAndPrint(char * data) {

```

```

    sscanf (data, "%3x%3x%3x", &red, &grn, &blu);
    char buffer[32];
    sprintf(buffer, "R%4.4d G%4.4d B%4.4d", red, grn, blu);
    Serial.println(buffer);
}

```

When using the ColorPAL to detect the color change in pH test strips, wet the whole strip with solution and then cut the strip in two. Place the two pieces side by side, so that there is a larger area for the ColorPAL to sense. When testing very small objects, the ColorPAL may erroneously include the color of the background behind. This will produce inaccurate results.

The Serial Monitor window displays the red, green, and blue components of the reflected color. These components are expressed in decimal value, with three digits for each color. The format is:

RRRGGBBB

The values read by the sensor are considered “raw” and uncalibrated. To more accurately determine the color of an object, the sensor should be calibrated against black and white reference colors. This topic is beyond this how-to, but more information on this topic may be found in the documentation for the Color Matching Program, described below.

Using the Color Matching Program with the Arduino

The Color Matching Program for the ColorPAL (a free download from the ColorPAL product page) runs on Windows XP or later. It works by looking for the ColorPAL sensor communicating over an open serial port. Once the sensor has been detected, you may use the Color Matching Program to calibrate it for black and white levels. Passing the ColorPAL detector window over an object visually shows you the color of that object.

To use the Color Matching Program with the Arduino, you must first change the waitDelay constant in the example sketch to increase the speed of the return values:

```
const int waitDelay = 10;    // Read once every 10 milliseconds
```

Upload the sketch to the Arduino, open the Serial Monitor window to start communication, wait about 5 seconds for the ColorPAL to reset and begin transmitting back color value, then close the window.

Start the Color Matching Program. The program should be able to find the serial port in use by the ColorPAL, and begin reading the value. If the Color Matching Program cannot find the ColorPAL on any serial port, repeat the above steps again. Occasionally it takes several tries to successfully establish communication.